
1. „Ciśniemy”, czyli przycisk

W tym odcinku wprowadzimy kolejną modyfikację do naszego układu. Tym razem
użyjemy  przycisku  aby  włączyć  i  wyłączyć  diodę  LED.  Samo  zagadnienie  jest  banalne,
ciekawsze  są jak zwykle „okoliczności przyrody”. Chociaż na rynku dostępne są różnorodne
przyciski, użyjemy tutaj przycisku (przełącznika) monostabilnego typu Tact Switch. Tak więc
do listy naszych potrzeb dopiszemy:

1 przycisk typu Tact Switch około 0,5 PLN

Cena jest orientacyjna, wg najtańszych ofert na Popularnym Portalu Aukcyjnym1.

Przycisk  jest  szczególnym  rodzajem  przełącznika,  wydzielonym  ze  względu  na
swoją budowę, a przede wszystkim sposób uruchomienia – naciśnięcie (a nie np. przesunięcie
dźwigni). Przycisk monostabilny to taki rodzaj przycisku który może zmienić swój stan na
czas naciśnięcia,  po czym wraca do stanu poprzedniego.  W przypadku naszego przycisku
domyślnym stanem jest stan OFF (wyłączony, przez przycisk nie przepływa prąd), który na
czas  naciśnięcia  zmienia  się  na  ON (włączony,  prąd  przepływa).  Istnieją  także  przyciski
bistabilne, tj zmieniające trwale stan, do czasu kolejnego naciśnięcia. Takie przyciski także
nadają się do naszego zadania, gdyż badać będziemy stan przycisku a nie sposób jego zmiany.

    

Przycisk i jego symbol Sposób działania  przycisku monostabilnego

1http://www.allegro.pl/  

http://www.allegro.pl/


Nasz przycisk ma 4 nóżki  (piny),  co może być nieco  mylące,  np.  sugerować,  iż
obsługuje  dwa  obwody.  W  rzeczywistości  przeciwległe  nóżki  są  ze  sobą  wewnętrznie
połączone,  a  taka  budowa  jest  podyktowana  względami  stabilności  montażu  przycisku.
Wewnątrz  znajduje się  także sprężynka,  która powoduje powrót  do stanu domyślnego po
zwolnieniu przycisku. Wybrałem przycisk o wymiarach 12×12 mm, którego nóżki (z pewnym
trudem) da się umieścić w płytce stykowej2. Generalnie przełączniki (więc i przyciski) tego
typu określane są skrótem SPST-NO3.

Aby przygotować obwód realizujący nasze zagadnienie nie jest potrzebny mikrokon-
troler.  Wystarczy  odpowiednio  podłączyć  diodę  LED  przez  rezystor  z  jednej  strony  do
zasilania,  z  drugiej  poprzez  przycisk  do  masy.  Wówczas  naciśnięcie  przycisku  będzie
zamykać  obwód  i  dioda  będzie  świecić.  Natomiast  naszym  celem  jest  wykorzystanie
mikrokontrolera  do  tego  zadania.  Można  to  zrealizować  podłączając  przycisk  do  jednej
z nóżek mikrokontrolera, którą ustawimy jako wejście a diodę LED do innej nóżki ustawionej
jako wyjście. Zastanówmy się teraz jak to zrealizować w praktyce.

Mikrokontrolery AVR radzą sobie dobrze z określaniem stanu wysokiego i niskiego
na danej linii wejścia-wyjścia, robiąc to w prosty sposób: napięcie poniżej połowy napięcia
zasilania  jest  uznawane za  stan  niski  (cyfrowe 0),  należące  do  drugiej  połowy – za  stan
wysoki  (cyfrowe  1).  Zatem należałoby  za  pomocą  przycisku  spowodować  na  danej  linii
wejściowej zmianę napięcia ze stanu wysokiego na niski i z powrotem. Intuicja podpowiada,
że można połączyć odpowiednią nóżkę albo z napięciem zasilania, co teoretycznie powinno
wymusić na pinie stan wysoki, albo z masą, co powinno skutkować stanem niskim na tej
nóżce. 

Przyjrzyjmy się drugiej ewentualności. Naciśnięcie przycisku wymusi stan niski na
linii  wejściowej,  ale  jaki  będzie  stan  linii  po  zwolnieniu  przycisku?  Będzie  niestabilny
i   podobnie  jak  niepodłączony przewód,  będzie  działał  jak  antena  nieustannie  zmieniając
napięcie  między stanem niskim a wysokim. Takie działanie  jest  niedopuszczalne i  należy
zrobić coś co ustabilizuje linię w stanie wysokim w czasie gdy przycisk jest zwolniony. 

2 Trudno znaleźć  informację  o  producencie,  ale  z  pewnością  to  przycisk  produkowany w ChRL.  Wg noty
sprzedawcy maksymalne napięcie pracy przełącznika to 12 V a prąd 50 mA.
3 Nazwa  ta  dotyczy  sposobu  działania  przełącznika.  Pierwsze  dwie  litery  dotyczą  liczby  obwodów,  które
obsługuje przełącznik. Może to być SP (single pole  - „jeden biegun” czyli jeden obwód), DP (double pole -
„podwójny biegun” - obsługuje dwa obwody), 3P i 4P. Kolejne dwie litery dotyczą liczby pozycji zwiernych, np.
ST (single throw – łączy obwód tylko na jednej z pozycji), DT (double throw – łączy obwód w dwóch pozycjach,
np. jako przełącznik z zasilania sieciowego na bateryjne). Ostatni skrót to NO (normally open, domyślnie obwód
otwarty) lub NC (normally closed, domyślnie obwód zamknięty). 

Najprostsze  rozwiązanie  to  połączenie  nóżki  linii  wejściowej  mikrokontrolera  do
zasilania, jednak skończyłoby się to katastrofalnie po naciśnięciu przycisku. Po bezpośrednim
połączeniu napięcia zasilającego z masą niechybnie powstałoby zwarcie w układzie. Ale jest
na to rada: rezystor podciągający (pull up) do zasilania. 

Rezystor podciągający to rezystor o względnie wysokiej oporności (od 10 kΩ do 100
kΩ), który w czasie gdy przycisk nie jest naciśnięty podciąga linię wejściową do zasilania,
inaczej  mówiąc  określa  domyślną  wartość  napięcia  na  nóżce.  Po  naciśnięciu  przycisku
tworzone jest  połączenie z  masą o bardzo niskiej  oporności  w porównaniu do połączenia
z zasilaniem przez rezystor, dzięki czemu przez ten rezystor popłynie tylko niewielki prąd.
Jednocześnie napięcie na nóżce mikrokontrolera będzie bliskie 0 V. Rozwiązanie z rezystorem
wygląda więc obiecująco.

Jest jeszcze jedna dobra wiadomość: w mikrokontrolerach AVR każda linia wejścia-
wyjścia  ma  dodatkowo  własny  rezystor  podciągający  i  nie  ma  potrzeby  dołączania
dodatkowego rezystora do układu. Należy jedynie aktywować rezystor wewnętrzny na danym
pinie i poprzez naciśnięcie przycisku na linii tej pojawi się stan niski, który po zwolnieniu
przycisku  wróci  do  poziomu  wysokiego.  Aktywowanie  wewnętrznego  rezystora
podciągającego to już jednak sprawa kodu.

2. Układ

Po tym przydługim wstępie wróćmy do naszego układu. Na pewno część układu
musimy  zdemontować,  gdyż  nie  będziemy  potrzebować  trzech  diod  LED  oraz  trzech
rezystorów,  wystarczy  nam  jedna  dioda  i  jeden  rezystor.  Pozostawimy  więc  diodę  LED
podpiętą  do  pinu  5  (PB0)  mikrokontrolera  oraz  rezystor  220  Ω łączący  ją  z  linią  masy.
Dodamy przycisk połączony z nóżką 6 (PB1) w miejscu jednej z usuniętych diod LED.

Dla  przypomnienia  -  podczas  programowania  mikrokontrolera  przy  pomocy
programatora USBasp linia masy diody LED oraz przycisku jest odłączona więc nie mogą
one przeszkadzać w zaprogramowaniu mikrokontrolera. Natomiast podczas normalnej pracy
mikrokontrolera dioda LED połączona z jego nóżką 5 powinna się zaświecić po naciśnięciu
przycisku połączonego z pinem 6 i zgasnąć po zwolnieniu przycisku.

3. Kod

Nasz  kod  źródłowy  powinien  pozwolić  zaprogramować  mikrokontroler  w  taki
sposób  aby  ten  mógł  zrealizować  dwa  zadania:  1)  programowe  wykrycie  naciśnięcia  i
zwolnienia przycisku; 2) reakcję na te zdarzenia poprzez zaświecenie i zgaszenie diody LED. 

Aby zrealizować pierwsze zadanie możemy wykorzystać część kodu z poprzedniego
odcinka, tj. pozostawić definicje stałych PB0 (maska bitowa nóżki 5) oraz PB1 (maska bitowa
nóżki 6). Różnice pojawią się  w ustawieniu rejestru kierunków: do określenia pinu PB0 jako
wyjścia możemy użyć operacji OR „zapalając” dopowiadający mu bit, natomiast aby pin PB1
ustawić jako wejście musimy „zgasić” jego bit, używając np. operacji AND NOT. Ponadto
powinniśmy włączyć rezystor podciągający na PB1. Aby to zrobić musimy odwołać się do
rejestru PORTB „zapalając” bit PB14. 

Drugie  zadanie  zrealizujemy  w  pętli  nieskończonej.  W  każdym  przebiegu  pętli
powinniśmy  sprawdzać  czy  bit  odpowiadający  nóżce  6  mikrokontrolera  (PB1)  jest
„zgaszony”,  tzn  czy  na  tym pinie  mamy stan  niski  odpowiadający  naciśnięciu  przycisku.
Można tego dokonać np. sprawdzając wynik operacji AND na zawartości rejestru wejścia.
Rejestrem takim w naszym mikrokontrolerze jest PINB. Wynik koniunkcji bitowej wyniesie 1
jeśli bit PB1 będzie w stanie wysokim, 0 – jeśli będzie w stanie niskim. W zależności od tego
wyniku powinniśmy zapalać bądź gasić diodę LED na PB0.

Nasz kod może wyglądać następująco:

4 Użycie rejestru PORTB do włączenia rezystora podciągającego pinu ustawionego w trybie wejścia wydaje się
nieco mylące, wszak ten sam rejestr służy do obsługi pinów w trybie wyjścia, ustawiając na nich stan wysoki
(„zapalając” bit) lub niski („gasząc” bit). Można przypuszczać, że projektantom mikrokontrolerów AVR chodziło
o zaoszczędzenie miejsca w mikrokontrolerze poprzez użycie jednego rejestru do dwóch celów. O który cel
chodzi wskazuje rejestr kierunku DDRB.

program button;

{$IFNDEF attiny13}
  {$Fatal Invalid controler type, expected: attiny13}
{$ENDIF}

const
  PB0 = %00000001; // pin PB0 = nóżka 5 mikrokontrolera
  PB1 = %00000010; // pin PB1 = nóżka 6 mikrokontrolera

begin
  //rejestr kierunków
  DDRB := DDRB or PB0;           // pin PB0 jako wyjscie
  DDRB := DDRB and not PB1;      // pin PB1 jako wejscie

  PORTB := PORTB or PB1;         // włączenie rezystora podciągającego na 
PB1

  while true do                  // pętla nieskończona
  begin
    if (PINB and PB1) = 0 then   // jeśli przycisk został naciśnięty 
(bit=0)
      PORTB:=PORTB or PB0        // włącz diodę LED
    else                         // w przeciwnym wypadku
      PORTB:=PORTB and not PB0;  // wyłącz diodę LED
  end;
end.

Trzeba mieć na uwadze, że domyślnie wszystkie piny mikrokontrolera są ustawione
jako wejścia, tzn. wszystkie bity rejestru DDRB są ustawione jako 0. Jednocześnie wszystkie
bity rejestru PORTB również mają domyślnie wartość 0, co oznacza, że żaden wewnętrzny
rezystor podciągający nie jest włączony, tzn. wejścia nie są podciągnięte do zasilania. Warto
je  programowo  włączyć  aby  uchronić  się  przed  niestabilnym  napięciem  na  pinach
wejściowych, szczególnie w sytuacji kiedy badamy ich stan.

4. Zmiana stanu

Zasadniczo nasze zadanie zostało wykonane. Aby nie kończyć odcinka tak szybko,
można  nieco  je  urozmaicić  i  przystosować  do  bardziej  „życiowych”  zastosowań.  Często
przycisk potrzebny jest aby trwale zmienić stan jakiegoś urządzenia. Na przykład włączając
telewizor chcemy aby pozostał włączony także po zwolnieniu przycisku i został wyłączony
dopiero  po  ponownym jego  naciśnięciu.  Zmodyfikujmy  więc  nasze  zadanie  następująco:
dioda LED ma zaświecić po naciśnięciu przycisku i zgasnąć po ponownym jego naciśnięciu5.

Układ  możemy  pozostawić  bez  zmian,  musimy  jednak  zmienić  kod  tak,  aby
wykrywał zmianę stanu przycisku a nie tylko sam fakt, że przycisk został naciśnięty bądź nie.
Ponadto program musi wiedzieć czy aktualnie dioda świeci czy nie. Do tych celów możemy
użyć dwóch zmiennych typu Boolean, które nazwiemy ButtonPressed i  LedOn i które mogą
przyjmować dwie wartości:  true (prawda) i  false (fałsz). Zmienna  ButtonPressed zapamięta
aktualny stan przycisku oraz pomocna będzie w stwierdzeniu czy przycisk został  właśnie
naciśnięty. Zmienna  LedOn z kolei zapamięta stan diody LED tak, aby można było łatwo ją
zmienić na przeciwny używając negacji (NOT). Nasz kod może wyglądać następująco:

5 Czyli w taki sposób aby program reagował na naciśnięcie przycisku tak, jakby był przyciskiem bistabilnym.
Siłą rzeczy dalszy nasz kod nie będzie nadawał się do przycisków bistabilnych.

program button2;

{$IFNDEF attiny13}
 {$Fatal Invalid controler type, expected: attiny13}
{$ENDIF}

const
  PB0 = %00000001; // pin PB0 = nóżka 5 mikrokontrolera
  PB1 = %00000010; // pin PB1 = nóżka 6 mikrokontrolera

var
  ButtonPressed: boolean;
  LedOn: boolean;

begin
  //rejestr kierunków
  DDRB := DDRB or PB0;               // pin PB0 jako wyjscie
  DDRB := DDRB and not PB1;          // pin PB1 jako wejscie

  PORTB := PORTB or PB1;             // włączenie rezystora podciągającego 
na PB1

  ButtonPressed := false;
  LedOn := false;

  while true do                      // pętla nieskończona
  begin
    if (PINB and PB1) = 0 then       // jeśli przycisk został naciśnięty 
(bit=0)
    begin
      if ButtonPressed = false then
      begin
        ButtonPressed := true;

        LedOn := not LedOn;            // ustaw flagę na przeciwną
        if LedOn then
          PORTB := PORTB or PB0        // włącz diodę LED
        else
          PORTB := PORTB and not PB0;  // wyłącz diodę LED
      end
    end
    else
      ButtonPressed := false;
  end;
end.

Większość zmian w kodzie skupiona jest na zdarzeniu zmiany stanu przycisku na
naciśnięty poprzez badanie wartości zmiennej ButtonPressed. Po zmianie wartości z false na
true zmieniamy wartość zmiennej LedOn na przeciwną i w zależności od tego włączamy bądź
wyłączamy diodę LED.

Zasadniczo nasza zmiana  powinna działać.  Cóż...  Właściwie… prawie działa.  Po
krótkiej  zabawie  przyciskiem  w  pewnym  momencie  dioda  zaczyna  reagować  dziwnie,
zamiast zaświecać się – gaśnie, jak się wydaje w losowych momentach. Mi wystarczy zwykle
mniej  niż  10  prób  by  taka  sytuacja  zaistniała.  O  co  chodzi?  Wygląda  na  to,  że  trzeba

sprawdzić co faktycznie się dzieje podczas przyciskania i zwalniania przycisku, sięgając po
oscyloskop6.

W większości  przypadków oscylogram wyglądał  zupełnie  w porządku i  wahania
poziomu napięcia dawały przebieg prostokątny zgodny z naciśnięciami przycisku. Ale kilka
razy pojawiły się dziwne piki lub niestabilne, wahające się poziomy napięcia. 

Czas  niestabilności  na  moich  oscylogramach  wyniósł  od  około  3  do  5  ms.  Ich
przyczyna jest  prozaiczna:  powierzchnie styków przycisku nie  są idealnie  gładkie i  przez
krótką chwilę napięcie na nich waha się między stanem wysokim a niskim, jednak nasz kod
nie  potrafi  tego  wychwycić  i  interpretuje  każdy  stan  niski  jako  naciśnięcie  przycisku.

6 Teoretycznie  do  zobrazowania  problemu  wystarczyłby  analizator  stanów  logicznych  oraz  dobry  program
analizujący zmienność napięcia w czasie, np. Saleae Logic. Jest to narzędzie znacznie tańsze od oscyloskopu.

Naturalnym rozwiązaniem jest  sprawdzenie czy po chwili  stan przycisku będzie taki  sam
i w zależności od tego podejmowanie dalszych decyzji7.

Musimy  więc  wrócić  do  naszej  procedury  opóźniającej  HalfSecondDelay
modyfikując ją w taki sposób aby mogła realizować różne opóźnienia, począwszy od kilku
ms. Abyśmy mogli skorzystać z tej procedury bez zmiany jej logiki musimy pójść na pewien
kompromis.  Nasz licznik jest  16-bitowy więc jego maksymalna wartość to 65535, co jest
wartością niewiele większą niż wartość dla połowy sekundy (60000), co wiemy z poprzedniej
części kursu.  Możemy więc przyjąć,  że górny zakres wystarczająco poprawnego działania
naszej  procedury  to  500  ms.  Jaki  będzie  zakres  dolny?  1  ms  to  dla  naszego  licznika
120000 / 1000 czyli 120 a licznik jest zmniejszany o 1 co około 10 cykli mikrokontrolera.
Pamiętajmy, że przed właściwą pętlą wykonujemy operację mnożenia aby uzyskać wartość
początkową  licznika  przy  podanej  liczbie  milisekund.   Ta  operacja  również  zajmie  czas
mikrokontrolera, w zależności od tego jaki kod asemblera wygeneruje Free Pascal Compiler8.
Można oszacować, że będzie to kilkadziesiąt cykli, a dodając pewien zapas  - blisko 1 ms.
Oznacza to, że nasza procedura będzie działała w zakresie 1-500 ms, jednak jej czas działania
może być dłuższy nawet o 1 ms9. Naszą nową procedurę opóźniającą możemy napisać tak:

procedure DelayMs(Value: UInt16);
var
  wCounter: UInt16;
  bHi, bLo: byte;
label
  Delay;
begin
  wCounter:= Value * 120; // kilkadziesiąt cykli zegarowych
  bHi:= Hi(wCounter);     // kilkanaście cykli
  bLo:= Lo(wCounter);     // zegarowych
  asm
    ldd R25, bHi // 2 cykle zegarowe
    ldd R24, bLo  // 2 cykle zegarowe
  Delay:
    sbiw R24,1   // 2 cykle; zmniejsz licznik 16-bitowy o 1
    nop          // 1 cykl
    nop          // 1 cykl
    nop          // 1 cykl
    nop          // 1 cykl
    nop          // 1 cykl
    nop          // 1 cykl
    brne Delay   // 2 cykle jeśli skok do etykiety, 1 cykl jeśli nie
    nop          // 1 cykl
  end['r24','r25'];
end;

Z pewnością ta procedura nie jest bardzo dokładna, ale do naszych celów wcale nie
musi taka być. Jeśli konserwatywnie ustawimy parametr wejściowy na 10 ms (nawet jeśli
w rzeczywistości będzie to 11 czy 12 ms) to mało prawdopodobne aby jakikolwiek człowiek

7 Inny sposób – stosowany przed erą mikrokontrolerów to dodanie kondensatora wygładzającego. Współcześnie
jednak takie rozwiązania uchodzą za drogie (wszak kondensator niewiele, ale kosztuje) i taniej dopisać kilka
linijek kodu rozwiązując problem. 
8 Zagadnienie nie jest trywialne, gdyż ATTiny13 nie posiada sprzętowego wsparcia operacji mnożenia (MUL),
więc rozwiązanie musi być oparte np. na przesunięciach bitowych. 
9 Naturalnie można udoskonalić naszą procedurę odejmując od licznika liczbę cykli jaką zajmie operacja mno-
żenia, ale na nasze potrzeby możemy na razie zostawić to miejsce w kodzie bez zmian.

mógł  przed  tym  czasem  nacisnąć  przycisk  po  raz  drugi.  Tak  więc  naszą  główną  pętlę
programu możemy tak zmodyfikować, aby dokonać ponownego sprawdzenia czy przycisk
jest nadal naciśnięty po 10 ms:

  while true do                      // pętla nieskończona
  begin
    if (PINB and PB1) = 0 then       // jeśli przycisk został naciśnięty 
(bit=0)
    begin
      DelayMs(10);                   // konwerwatywnie, 10 ms opóźnienia
      if (PINB and PB1) = 0 then     // ponownie sprawdzamy przycisk
      begin
        if ButtonPressed = false then
        begin
          ButtonPressed := true;

          LedOn := not LedOn;            // ustaw flagę na przeciwną
          if LedOn then
            PORTB := PORTB or PB0        // włącz diodę LED
          else
            PORTB := PORTB and not PB0;  // wyłącz diodę LED
        end;
      end
    end
    else
      ButtonPressed := false;
  end;

Tak,  teraz  wyeliminowaliśmy  problem  drgań  styków  przycisku.  I  na  tym
moglibyśmy zakończyć odcinek, gdyby nie fakt, że nasz kod nie jest zbyt czytelny. Dużo
w nim operacji na bitach, są też maski bitowe. Może dałoby się go jakoś uprościć?

5. Uproszczenie kodu i pierwsza biblioteka

Cóż,  z procedurą  DelayMs niewiele  można zrobić poza dodaniem jeszcze więcej
asemblera  aby  doprecyzować  opóźnienie  wynikające  z  mnożenia  (jak  dotąd  tylko  je
szacujemy). Zostawimy ją teraz bez zmian. Ale czym zastąpić operacje bitowe? Otóż można
zastosować własne typy danych odpowiadające rozmiarowi bajtu, ale dające łatwiejszy dostęp
do bitów (por. odcinek 0 kursu):

type
  TBinary = 0..1; //dopuszczalne wartości to 0 i 1

  TBitRec = bitpacked record //deklaracja bajtu jako rekordu z polami 
reprezentującymi bity 0-7
    Bit0: TBinary;
    Bit1: TBinary;
    Bit2: TBinary;
    Bit3: TBinary;
    Bit4: TBinary;
    Bit5: TBinary;
    Bit6: TBinary;
    Bit7: TBinary;
  end;

Zdefiniowaliśmy w ten sposób dwa własne typy: TBinary, który przyjmuje wartości
0 lub 1 oraz TBitRec – rekord (strukturę) składającą się z ośmiu pól typu TBinary, o nazwach

odpowiadających bitom od 0 do 710.  TBitRec to po prostu bajt składający się z pól – bitów.
Teraz możemy go użyć w deklaracji zmiennych, w których potrzebujemy dostępu do bitów.
Free  Pascal  pozwala  na  rzutowanie  zmiennych  na  rejestry  za  pomocą  słowa kluczowego
absolute, więc zadeklarujemy zmienne DDRBits, PORTBBits oraz PINBBits tak, aby miały
te same adresy w pamięci jak rejestry odpowiednio DDRB, PORTB i PINB. W ten sposób
będziemy mieć prosty dostęp do bitów rejestrów, z których korzystamy:

var
  ButtonPressed: boolean;
  LedOn: boolean;

  DDRBBits: TBitRec absolute DDRB;
  PORTBBits: TBitRec absolute PORTB;
  PINBBits: TBitRec absolute PINB;
begin
  //rejestr kierunków
  DDRBBits.Bit0 :=  1;                 // PB0 jako wyjscie
  DDRBBits.Bit1 :=  0;                 // PB1 jako wejscie

  PORTBBits.Bit1 := 1;                 // włączenie rezystora 
podciągającego na PB1

  ButtonPressed := false;
  LedOn := false;

  while true do                        // pętla nieskończona
  begin
    if PINBBits.Bit1 = 0 then          // jeśli przycisk został naciśnięty 
(bit=0)
    begin
      DelayMs(10);                     // konserwatywnie, 10 ms opóźnienia
      if PINBBits.Bit1 = 0 then        // ponownie sprawdzamy przycisk
      begin
        if ButtonPressed = false then
        begin
          ButtonPressed := true;

          LedOn := not LedOn;          // ustaw flagę na przeciwną
          if LedOn then
            PORTBBits.Bit0 := 1        // włącz diodę LED
          else
            PORTBBits.Bit0 := 0        // wyłącz diodę LED
        end;
      end
    end
    else
      ButtonPressed := false;
  end;
end.

Jak widać, zamiast bardziej skomplikowanych operacji bitowych na całych bajtach,
możemy użyć bezpośredniego porównania pojedynczego bitu, bądź przypisania mu wartości
0  lub  1.  Taki  kod  dużo  łatwiej  jest  czytać,  choć  w porównaniu  do  poprzedniego  różnic
funkcjonalnych nie ma.

Na koniec popatrzmy na kod z „dalszej” perspektywy. Jeśli będziemy pisać kolejne
programy  dla  ATTiny13,  z  pewnością  będziemy  wielokrotnie  korzystać  z  definicji  typu
10 Nazwy pól tego typu rekordowego mogą być dowolne, jednak musi być łącznie 8 i muszą być typu TBinary.

TBitRec ze względu na jego wygodę. Możemy także przypuszczać, że w przyszłości przyda
nam się procedura DelayMs. Warto więc wydzielić ich kod do nowego modułu, tworząc w ten
sposób pierwszą bibliotekę (na razie zawierającą jeden moduł). Nowy moduł możemy nazwać
attiny13_basics.pas, gdyż zawierać będzie podstawowe procedury i typy na tyle uniwersalne
że  mogą  się  przydać  w  innych  programach.  Moduł  musi  mieć  sekcję  interface,  gdzie
zadeklarujemy nasze typy i procedurę oraz  implementation, gdzie umieścimy kod procedury:

unit attiny13_basics;

{$IFNDEF attiny13}
 {$Fatal Invalid controler type, expected: attiny13}
{$ENDIF}

interface

type
  TBinary = 0..1; //dopuszczalne wartości to 0 i 1

  TBitRec = bitpacked record //deklaracja bajtu...
    //[...] tutaj deklaracja pól
  end;

procedure DelayMs(Value: UInt16);

implementation

procedure DelayMs(Value: UInt16);
var
  //[...] tutaj lista zmiennych
label
  //[...] tutaj lista etykiet
begin
  //[...] tutaj kod procedury
end;

end.

Nasz nowy moduł  zapiszmy w tym samym katalogu,  w którym zapisaliśmy plik
programu11. Aby skorzystać z niego w programie musimy dodać sekcję uses z listą modułów
wykorzystywanych w programie. Teraz struktura naszego programu będzie następująca: 

program button4;

{$IFNDEF attiny13}
 {$Fatal Invalid controler type, expected: attiny13}
{$ENDIF}

uses
  attiny13_basics;                    // lista wykorzystywanych modułów

var
  ButtonPressed: boolean;
  LedOn: boolean;
  
  // korzystamy z typów z modułu

11Można również zapisać plik modułu w innym katalogu, wówczas w programie należałoby wskazać ścieżkę
względną do modułu albo moduł zapisać do katalogu lib AVRPascala. Myślę jednak, że przy pisaniu pierwszych
modułów nie warto komplikować sobie życia...

  DDRBBits: TBitRec absolute DDRB;    
  PORTBBits: TBitRec absolute PORTB;
  PINBBits: TBitRec absolute PINB;
begin
  //[...] tutaj inicjalizacja bitów rejestrów i zmiennych 
  while true do                      // pętla nieskończona
  begin
    if PINBBits.Bit1 = 0 then          
    begin
      DelayMs(10);                   // korzystamy z procedury z modułu
      //[...] dalszy kod
    end
    //[...] dalszy kod
  end;
end.

W ten sposób dotarliśmy do końca trzeciego odcinka kursu. Odcinek nie wyczerpał
oczywiście  tematu,  gdyż  przedstawiłem  tylko  jeden  z  możliwych  sposobów  współpracy
układu z przyciskiem z wykorzystaniem rezystora podciągającego. Równie dobrze (?) można
zbudować układ niejako „na odwrót”, używając zewnętrznego rezystora ściągającego (pull-
down)12 i  połączyć  przycisk z  linią  zasilania.  Wymagałoby to jednak pewnej  przebudowy
naszego układu. Potencjał do optymalizacji ma także kod programu.

Życzę miłej zabawy!

Andrzej Karwowski

Nobody is perfect – znalazłeś/-aś błąd lub nieścisłość, masz pytanie bądź sugestię – napisz na 
adres ackarwow@gmail.com 

Podziękowania

Dziękuję użytkownikowi  sp3ots z forum  www.elektroda.pl za inspirację do przygotowania
niniejszego odcinka kursu. 

12 ATTiny13 nie posiada wewnętrznego rezystora pull-down.

